In[1]:= `<< Hyper.m`

In[2]:= `? Hyper`

Hyper[eqn, y[n]] finds at least one hypergeometric
solution of the homogeneous equation eqn over the field of rational numbers Q
(provided any such solution exists). Hyper[eqn, y[n], Solutions –> All] finds
a generating set (not necessarily linearly independent) for the space of
solutions generated by hypergeometric terms over Q. Hyper[eqn, y[n],
Quadratics –> True] finds solutions over quadratic extensions of Q. Solutions
y[n] are given by their rational representations y[n+1]/y[n].
Warning: The worst–case time complexity of Hyper is exponential
in the degrees of the leading and trailing coefficients of eqn.

In[3]:= `Hyper[(n - 1) y[n + 2] - (3 n - 2) y[n + 1] + 2 n y[n] == 0, y[n]]`

Out[3]= $\left\{\frac{1+n}{n}\right\}$

In[4]:= `Hyper[(n - 1) y[n + 2] - (3 n - 2) y[n + 1] + 2 n y[n], y[n]]`

Out[4]= $\left\{\frac{1+n}{n}\right\}$

In[5]:= `Hyper[(n - 1) y[n + 2] - (3 n - 2) y[n + 1] + 2 n y[n], y[n], Solutions → All]`

Out[5]= $\left\{2, \frac{1+n}{n}\right\}$

In[6]:= ``<< RISC`fastZeil` ``

Fast Zeilberger Package version 3.61
written by Peter Paule, Markus Schorn, and Axel Riese
Copyright Research Institute for Symbolic Computation (RISC),
Johannes Kepler University, Linz, Austria

We continue with the last example for Zeilberger's algorithm from GosperZeilberger.nb (related to non-minimality)

In[7]:=
```
d = 3;
rec = Zb[(-1)^k Binomial[n, k] Binomial[d k, n], {k, 0, n}, n]
```

```
If `n' is a natural number, then:
```

Out[8]= {9 (1 + n) (2 + n) SUM[n] + 3 (2 + n) (7 + 5 n) SUM[1 + n] + 2 (2 + n) (3 + 2 n) SUM[2 + n] ==
 -n (3 + 2 n) Binomial[0, n]}

As before, we extract the recurrence in a way that can be processed (Note, that the RHS is zero for all n≥ 0 as discussed last time)

In[9]:=
```
rec = ReleaseHold[rec][[1, 1]] /. SUM → a
```
Out[9]= $9\,(1+n)\,(2+n)\,a[n]+3\,(2+n)\,(7+5\,n)\,a[1+n]+2\,(2+n)\,(3+2\,n)\,a[2+n]$

In[10]:=
```
Hyper[rec, a[n], Solutions → All]
```
Out[10]= $\{-3\}$

So, $(-3)^n$ that we identified as the closed form of the sum the last time, is the only hypergeometric solution of this recurrence.

Irrationality of $\zeta(3)$ (an example taken from the wikipedia entry on Petkovšek’s algorithm)

In[11]:=
```
rec = Zb[Binomial[n, k]^2 Binomial[n + k, k]^2, {k, 0, n}, n]
```
```
If `n' is a natural number, then:
```
Out[11]= $\{(1+n)^3\,\mathrm{SUM}[n]-(3+2\,n)\,(39+51\,n+17\,n^2)\,\mathrm{SUM}[1+n]+(2+n)^3\,\mathrm{SUM}[2+n]==0\}$

In[12]:=
```
rec = ReleaseHold[rec][[1, 1]] /. SUM → a
```
Out[12]= $(1+n)^3\,a[n]-(3+2\,n)\,(39+51\,n+17\,n^2)\,a[1+n]+(2+n)^3\,a[2+n]$

In[13]:=
```
Hyper[rec, a[n], Solutions → All]
```
Out[13]= $\{\}$

this proves that $\sum\binom{n}{k}^2\binom{n+k}{k}^2$ has no hypergeometric solutions.

In[14]:=
```
<< RISC`HolonomicFunctions`
```

HolonomicFunctions Package version 1.7.3 (21-Mar-2017)
written by Christoph Koutschan
Copyright Research Institute for Symbolic Computation (RISC),
Johannes Kepler University, Linz, Austria

```
--> Type  ?HolonomicFunctions  for help.
```

In[15]:=
```
input = Sum[Binomial[d n, a n + 1], {d, 5}, {a, 2, 3}]
```
Out[15]=
```
Binomial[n, 1 + 2 n] + Binomial[n, 1 + 3 n] +
 Binomial[2 n, 1 + 3 n] + Binomial[3 n, 1 + 2 n] + Binomial[4 n, 1 + 2 n] +
 Binomial[4 n, 1 + 3 n] + Binomial[5 n, 1 + 2 n] + Binomial[5 n, 1 + 3 n]
```

In[16]:=
```
ann = Annihilator[input, S[n]] // Factor;
```

In[17]:=
```
Support[ann]
```
Out[17]= $\{\{S_n^8,\,S_n^7,\,S_n^6,\,S_n^5,\,S_n^4,\,S_n^3,\,S_n^2,\,S_n,\,1\}\}$

In[18]:=
```
Exponent[ann, n]
```
Out[18]= $\{108\}$

In[19]:=
```
data = Table[input, {n, 0, 500}];
```

In[20]:=
```
<< RISC`Guess`
```

Package GeneratingFunctions version 0.8 written by Christian Mallinger
Copyright Research Institute for Symbolic Computation (RISC),
Johannes Kepler University, Linz, Austria

Guess Package version 0.52
written by Manuel Kauers
Copyright Research Institute for Symbolic Computation (RISC),
Johannes Kepler University, Linz, Austria

In[21]:=
```
rec = GuessMinRE[data, a[n]];
```

In[22]:=
```
rec = Factor /@ rec;
```

In[23]:=
```
Cases[rec, a[_], Infinity] // Union
```

Out[23]= {a[n], a[1 + n], a[2 + n], a[3 + n], a[4 + n]}

In[24]:=
```
Exponent[rec, n]
```

Out[24]= 32

In[25]:=
```
TimeConstrained[Hyper[rec, a[n], Solutions → All], 30]
```

```
Warning: irreducible factors of degree > 1 in leading coefficient;
some solutions may not be found

Warning: irreducible factors of degree > 1 in trailing coefficient;
some solutions may not be found
```

Out[25]= $Aborted

In[26]:=
```
TimeConstrained[Hyper[rec, a[n]], 30]
```

```
Warning: irreducible factors of degree > 1 in leading coefficient;
some solutions may not be found

Warning: irreducible factors of degree > 1 in trailing coefficient;
some solutions may not be found
```

Out[26]= $\left\{\frac{3\,(1+3\,n)\,(2+3\,n)}{2\,n\,(3+2\,n)}\right\}$